

1. ATL Transformation Example: Make → Ant

The Make to Ant example describes a transformation from a Makefile to a file in Ant.

1.1. Transformation overview

The aim of this transformation is to generate a file for the build tool Ant starting from a Makefile.

```
#test of a makefile

CC=gcc
CFLAGS=-Wall -ansi
LDFLAGS=-Wall -ansi

hello : hello.o main.o
    $(CC) -o hello hello.o main.o $(LDFLAGS)
    @skip

hello.o : hello.c
    $(CC) -o hello.o -c hello.c $(CFLAGS)


main.o : main.c hello.h
    @$(CC) -o main.o -c main.c $(CFLAGS)

clean :
    rm -rf *.o

mrproper : clean
    rm -rf $(EXEC)
```

Figure 1. Example of Makefile

```
<project name="Hello">
  <description>Test d'un fichier makefile</description>
  <property name="CC" value="gcc"/>
  <property name="CFLAGS" value="-Wall -ansi"/>
  <property name="LDFLAGS" value="-Wall -ansi"/>
  <target name="hello" depends="hello.o,main.o">
    <echo message="$(CC) -o hello hello.o main.o $(LDFLAGS)"/>
    <exec executable="$(CC) -o hello hello.o main.o $(LDFLAGS)"/>
    <exec executable="skip"/>
  </target>
  <target name="hello.o" >
    <echo message="$(CC) -o hello.o -c hello.c $(CFLAGS)"/>
    <exec executable="$(CC) -o hello.o -c hello.c $(CFLAGS)"/>
  </target>
  <target name="main.o" >
    <exec executable="$(CC) -o main.o -c main.c $(CFLAGS)"/>
  </target>
  <target name="clean" >
    <echo message="rm -rf *.o"/>
    <exec executable="rm -rf *.o"/>
  </target>
  <target name="mrproper" depends="clean">
```

	ATL TRANSFORMATION EXAMPLE	
	Make to Ant	Date 05/08/2005

```

<echo message="rm -rf $(EXEC)"/>
<exec executable="rm -rf $(EXEC)"/>
</target>
</project>

```

Figure 2. The corresponding file in Ant




Figure 3. Transformation overview

This transformation is divided into several parts:

- the injector to obtain a file in file in xmi-format corresponding to the Make Metamodel;
- the transformation from the Make to the Ant Metamodel;
- the extractor to obtain a file in xml-format corresponding to Ant.

1.2. Metamodels

1.2.1. Make Metamodel




Figure 4. The Make metamodel

A Make is modeled by a Makefile element. This element contains several elements. An abstract Element can be either a Macro (to give a value to a constant) or a Rule.

A Macro has two attributes:

- the constant's name;
- the constant's value.

A Rule contains dependencies and shellLines (the commands which are called). There are two kinds of dependencies:

- the dependency of a file;

- the dependency of another rule.

A shellLine is a command which is called. When this command begins by the character '@', that means that there is no need to echo the command (the attribute *display* gets the value 'false').

1.2.2. Ant Metamodel




Figure 5. A simplified Ant Metamodel

An Ant file is modeled by a Project element. A project defines an ordered set of properties and one or more targets and which target is called by default.


A property serves to give a value to a constant.

A target is a set of tasks which must be executed after the execution of all its dependencies. For the make, there only two useful tasks:

- echo, its role is to echo a message to the current loggers;
- exec, its role is to execute a system command.

1.3. Injector

The injector is divided in several parts:

	ATL TRANSFORMATION EXAMPLE	
	Make to Ant	Date 05/08/2005

- transformation from the makefile to a file in xml-format;
- importation of the xml model, to obtain a file in xmi-format corresponding to the XML Metamodel;
- transformation from a file corresponding to the XML Metamodel to a file corresponding to the Make Metamodel.

1.3.1. Transformation from the Makefile to a file in xml-format

The first transformation consists in creating a xml-based file from the makefile.

```

<make>
  <comment>Test d'un fichier makefile</comment>
  <macro name="CC" value="gcc"/>
  <macro name="CFLAGS" value="-wall -ansi"/>
  <macro name="LDFLAGS" value="-wall -ansi"/>
  <rule name="hello" depends="hello.o main.o">
    <shellLine>@$(CC) -o hello hello.o main.o $(LDFLAGS)</shellLine>
    <shellLine>skip</shellLine>
  </rule>
  <rule name="hello.o" depends="hello.c">
    <shellLine>@$(CC) -o hello.o -c hello.c $(CFLAGS)</shellLine>
  </rule>
  <rule name="main.o" depends="main.c hello.h">
    <shellLine>$(CC) -o main.o -c main.c $(CFLAGS)</shellLine>
  </rule>
  <rule name="clean">
    <shellLine>rm -rf *.o</shellLine>
  </rule>
  <rule name="mrproper" depends="clean">
    <shellLine>@rm -rf $(EXEC)</shellLine>
  </rule>
</make>

```

Figure 6. XML-based file representing a Makefile

To obtain this file, the command Unix is used: sed.

The XML-based file begins by the tag '<make>' and finished by the tag '</make>'.


The comments are recognized by the character '#' at the beginning of the line. All characters after '#' are copied without analysis:

```
s/#\(.*\)$/ <comment>\1</comment>/ p
```

The macros are recognized thanks to the character '='. The possible space and tab characters which are placed before or after the character '=' or after the value of the Macro are deleted. In this analysis, it can not have comments in the same line.

```
s/^\([a-zA-Z][a-zA-Z0-9_]*\)[ ]*=[ ]*\([A-Za-z0-9_]*\)\)/ <macro name="\1" value="\2"/>/ p
```

The rules are recognized by the character ':' and a shell command is after a tab character. The possible space and tab characters which are placed before or after the character ':' or after the last dependency are deleted. But the spaces between two names of dependencies are not analysed. Concerning the shell command, all characters after the tab character are copied until the

	ATL TRANSFORMATION EXAMPLE	
	Make to Ant	Date 05/08/2005

last letter (or number) which is found in this line. In this analysis, it can not have comments inside a rule.

```

s/^\([A-Za-z][A-Za-z\.\0-9]*\) [ ]*:[ ]*\([A-Za-z0-9]\)\) [ ]*$/ <rule
name="\1" depends="\2">/
  p
  n
  :s
  /^ /{
    s/^ //
    s/\([A-Za-z0-9]\)\) [ ]*$/ <shellLine>\1</shellLine>/
    p
    n
    bs
  }
  a\
</rule>

```

The plug-in 'org.atl.eclipse.km3importer_1.0.0' creates a file in xmi.

1.3.2. Transformation from the XML Metamodel to the Make Metamodel

1.3.2.1. Rules Specification

These are the rules to transform a XML Model to a Make Model:

- For the Root, a Makefile element is created,
- For an Element which name is 'comment', a Comment element is created,
- For an Element which name is 'macro', a Macro element is created,
- Etc.

1.3.2.2. ATL Code

This ATL code for the XML to Make transformation consists of 7 helpers and 5 rules.

The getList helper is useful to extract the different names of dependencies for a rule. This list of names is given in parameter and a Sequence of String is returned. Two names of dependencies are separated by a space character. This helper uses another helper named getListAux.


The getAttribute helper is useful for all elements having Attribute children. Its rule consists in returning a value of an attribute whose name is given in parameter. It returns "" if the required attribute does not exist. This helper uses testAttribute helper which indicates whether the attribute given in parameter exists, and getAttrVal helper which returns the value of an attribute.

The getText helper is useful for all element having Text children. Its rule consists in returning the value of a text whose name is given in parameter. In this helper, there is no test of existence.

The getShellLine helper allows extracting a shell command: that is to say to remove the first character if this one is the character '@'.

The rule Root2Makefile allocates a Makefile element.

The rule Comment allocates a Comment element. ...

	ATL TRANSFORMATION EXAMPLE	
	Make to Ant	Date 05/08/2005

For the rule Rule, its dependencies can be represented by a FileDep or a RuleDep element. If it is a RuleDep element, a definition of this rule is present in this file:

```
itsRuleDep : Sequence(XML!Element) = XML!Element.allInstances() ->
  select(d | d.name = 'rule'
    and itsDependencies ->
      includes( d.getAttribute('name')));
```

If it is a FileDep element, there is no rule having its name:

```
itsFileDep : Sequence(String) = itsDependencies ->
  select(d | allRules ->
    select(e|e.getAttribute('name')=d)->
      isEmpty());
```

And the key words 'foreach [...] distinct' are used to create a RuleDep (or FileDep) element for all elements present in the constant itsRuleDep (or itsFileDep).

1.4. Transformation from Make to Ant

1.4.1. Rules Specification

These are the rules to transform a Make model to an Ant model:

- For the MakeFile element, a Project in Ant element is created. As there is not equivalent of 'default' in a Make, this the first rule found in the makefile which will be called by default.
- For each Macro, a Property is created.
- For each Rule, a Target is created. Only the dependencies of rules are kept as dependencies.
- There are two kinds of ShellLines:
 - When the attribute *display* is true, the tags *Echo* and *Exec* are created,
 - When the attribute *display* is false, only the tag *Exec* is created.

1.4.2. ATL Code

This ATL code for the Make to Ant transformation consists of 5 rules.

The rule Makefile2Project allocates a Project element. This element is given a name and a description. It is linked to the Macro and the Rule contained in the Makefile.

The rule Macro2Property allocates a PropertyValue element. This new Property is given a name and a value.


The rule Rule2Target allocates a Target element. This new Target is given a name and it contains Tasks and it is linked to others targets.

The rule ShellLine2Task_Display allocates two Tasks: Echo and Exec. The Task Echo id given a message and the task Exec is given a command.

The rule ShellLine2Task_NoDisplay allocates the Task Exec.

1.5. Extractor

The extractor is divided into several parts:

	ATL TRANSFORMATION EXAMPLE	
	Make to Ant	Date 05/08/2005

- Transformation from a file corresponding to the Ant Metamodel to a file corresponding to the XML Metamodel,
- Creation of the XML file in Ant.

1.5.1. Transformation from Ant to XML Metamodel

1.5.1.1. Rules Specification

These are the rules to transform an Ant Model to a XML Model:

- For the Project, a Root element is created,
- For a Comment element, an Element which name is 'comment' is created,
- Etc.

1.5.1.2. ATL Code

This ATL code for the Ant to XML transformation consists of 1 helper and 24 rules.

The concat helper allows concatenating a sequence of string given in parameter. Two elements are separated by a comma. This helper is useful for the attribute depends of a target.

The rule Project2Root creates a Root element for the projects having an attribute named description:

```
rule Project2Root{
  from i : Ant!Project(
    if i.description.oclIsUndefined()
    then false
    else not(i.description='')
    endif
  )
  to o : XML!Root(...)
}
```


The 'if then else' instruction is used: when the first test failed, the second is not executed.

There is another rule Project2RootWithoutDescription for the project not having description. Thus, there is no Attribute element named 'description' which has no value.

There is a rule per element.

I. Make metamodel in KM3 format

```
1 package Make {
2   --@begin central Class
3   class Makefile {
4     attribute name : String;
5     reference comment[0-1] container : Comment;
6     reference elements[1-*] ordered container : Element;
7   }
8   --@end central Class
9
10  --@begin Elements
11  --@comments represents differents elements contained in the makefile
12  abstract class Element {
13    attribute name : String;
14  }
15
16  --@comments represents a rule : a group of dependencies and of shellLines
17  class Rule extends Element {
18    reference dependencies[*] ordered container : Dependency;
19    reference shellLines [1-*] ordered container : ShellLine oppositeOf
20  ruleShellLine;
21  }
22
23  --@comments represents a macro : to give a value
24  class Macro extends Element {
25    attribute value : String;
26  }
27  --@end Elements
28
29  --@begin shellLine
30  class ShellLine{
31    attribute command : String;
32    attribute display : Boolean;
33    reference ruleShellLine : Rule oppositeOf shellLines;
34  }
35  --@end shellLine
36  --@begin comment
37  class Comment{
38    attribute text : String;
39  }
40  --@end comment
41  --@begin dependencies
42  --@comments represents a dependency contained in a rule
43  abstract class Dependency {
44  }
45
46  --@comments represents a dependency which call another rule
47  class RuleDep extends Dependency {
48    reference ruledep : Rule;
49  }
50
51  --@comments represents a file dependency
52  class FileDep extends Dependency {
53    attribute name : String;
54  }
55  --@end dependencies
56 }
57
58 package PrimitiveTypes{
```

	ATL TRANSFORMATION EXAMPLE	
	Make to Ant	Date 05/08/2005

```
59     datatype String;  
60     datatype Boolean;  
61 }
```

II. Ant metamodel in KM3 format (excerpt)


```
1  package Ant{
2
3    class Project{
4      attribute name : String;
5      attribute description [0-1] : String;
6      reference properties [*] ordered container : Property;
7      reference targets [1-*] container : Target;
8      reference "default" : Target;
9    }
10
11   abstract class Property{}
12
13   abstract class PropertyName extends Property{
14     attribute name : String;
15   }
16
17   class PropertyValue extends PropertyName{
18     attribute value : String;
19   }
20
21   class Target{
22     attribute name : String;
23     reference depends [*] : Target;
24     reference tasks [*] ordered container : PredefinedTask oppositeOf target;
25   }
26
27   abstract class Task{
28   }
29   abstract class PreDefinedTask extends Task{
30     reference target : Target oppositeOf tasks;
31   }
32
33   abstract class ExecutionTask extends PreDefinedTask{}
34
35   class Exec extends ExecutionTask{
36     attribute executable : String;
37     attribute dir : String;
38   }
39
40   abstract class MiscellaneousTask extends PreDefinedTask{}
41
42   class Echo extends MiscellaneousTask{
43     attribute message : String;
44   }
45 }
46 package PrimitiveTypes{
47   datatype String;
48 }
```

III. Make2Ant.sed file

```

1  #!/bin/sed -nf
2
3  # sed -nf make2xml.sed makefile >makefile.xml
4
5  li\
6  <make>
7
8  :d
9  /\$/{
10     s/[ ]*\$//
11     N
12     s/\n[ ]*/ /
13     bd
14 }
15
16 # Get rid of comments
17 s/#\(.*\)$/ <comment>\1</comment>/ p
18
19 /^[a-zA-Z][a-zA-Z0-9_]*[ ]*={
20     s/^\([a-zA-Z][a-zA-Z0-9_]*\)[ ]*=[ ]*\([A-Za-z0-9]\)\)[ ]*$/ <macro
21     name="\1" value="\2"/>/ p
22 }
23
24 #rule with dependencies
25 /^[A-Za-z][A-Za-z\0-9]*[ ]*:[ ]*\([A-Za-z].*\)$/{
26     <rule name="\1">\n <dependencies>\2</dependencies>/
27     s/^\([A-Za-z][A-Za-z\0-9]*\)[ ]*:[ ]*\([A-Za-z0-9]\)\)[ ]*$/ <rule name="\1"
28     depends="\2">/
29     p
30     n
31     :s
32     /^ /{
33         s/^ //
34         s/\([A-Za-z0-9]\)\)[ ]*$/ <shellLine>\1</shellLine>/
35         p
36         n
37         bs
38     }
39     a\
40     </rule>
41 }
42
43 #rule without dependencies
44 /^[A-Za-z][A-Za-z\0-9]*[ ]*:[ ]*$/{
45     <rule name="\1">\n <dependencies>\2</dependencies>/
46     s/^\([A-Za-z][A-Za-z\0-9]*\)[ ]*:[ ]*$/ <rule name="\1">/
47     p
48     n
49     :s
50     /^ /{
51         s/^ //
52         s/\([A-Za-z0-9]\)\)[ ]*$/ <shellLine>\1</shellLine>/
53         p
54         n
55         bs
56     }
57     a\
58     </rule>

```

	ATL TRANSFORMATION EXAMPLE	
	Make to Ant	Date 05/08/2005

59 }
60
61 \$a\
62 </make>

IV. XML2Make.atl file

```
1  module XML2Make;
2  create OUT : Make from IN : XML;
3
4
5  -- -- to extract a list of String from a String -- --
6
7  -- helper getList: extract a sequence of String from the String listString
8  -- in the same order
9  -- (two elements are separated by a comma)
10 helper def:getList(listString : String):Sequence(String)=
11     if(listString.size()==0)
12         then Sequence{}
13     else thisModule.getListAux(listString,1,1,Sequence{})
14     endif;
15
16 -- helper getListAux
17 -- index1: begin of the word
18 -- index2: meter
19 helper def:getListAux(listString: String, index1: Integer, index2: Integer,
20 provSequence: Sequence(String)): Sequence(String)=
21     if (listString.size()<index2)
22         then provSequence -> append(listString.substring(index1,index2-1))
23     else
24         if listString.substring(index2,index2)=' '
25             then thisModule.
26                 getListAux(listString,index2+1,index2+1,provSequence ->
27                     append(listString.substring(index1,index2-1)))
28             else thisModule.
29                 getListAux(listString,index1,index2+1, provSequence)
30             endif
31         endif;
32
33 -- -- to get an attribute -- --
34
35 -- helper getAttrVal: returns the value of the attribute 'name'
36 -- (without test of existence)
37 helper context XML!Element def: getAttrVal(name: String) : String =
38     self.children ->
39         select(c | c.ocIsKindOf(XML!Attribute) and c.name = name)->
40             first().value;
41
42 -- helper testAttribute: returns true if the attribute 'name' is defined
43 helper context XML!Element def: testAttribute(name : String) : Boolean =
44     not (self.children ->
45         select(d | d.ocIsKindOf(XML!Attribute) and d.name = name)->
46             first().ocIsUndefined());
47
48 -- helper: getAttribute: returns the value of the attribute
49 -- given in parameter
50 -- returns '' if this attribute does not exist
51 helper context XML!Element def:getAttribute(name : String):String =
52     if (self.testAttribute(name))
53         then self.getAttrVal(name)
54     else ''
55     endif;
56
57 -- -- others helpers -- --
58
```

```
59 -- helper getText:returns the value of a text belonging to an element
60 helper context XML!Element def: getText() : String =
61     self.children ->
62         select(c | c.ocIsKindOf(XML!Text)) ->
63             first().value;
64
65 -- helper getShellLine: deletes the first character if this one is '@'
66 helper context XML!Element def : getShellLine():String=
67     let thisText : String = self.getText() in
68     if (thisText.substring(1,1)='@')
69         then thisText.substring(2,thisText.size())
70     else thisText
71     endif;
72
73 -- rules
74
75 -- central rule
76 rule Root2MakeFile{
77     from i : XML!Root
78     to o : Make!Makefile(
79         name <- 'makefile',
80         comment <- i.children ->
81             select(d | d.ocIsKindOf(XML!Element) and d.name = 'comment')->
82                 first(),
83         elements <- i.children ->
84             select(d | d.ocIsKindOf(XML!Element)
85                 and not (d.name = 'comment'))
86     )
87 }
88
89 rule Comment{
90     from i : XML!Element(
91         i.name = 'comment'
92     )
93     to o : Make!Comment(
94         text <- i.getText()
95     )
96 }
97
98 rule Rule{
99     from i : XML!Element(
100         i.name = 'rule'
101     )
102     using{
103         allRules : Sequence(XML!Element) = XML!Element.allInstances() ->
104             select(d | d.name = 'rule' );
105         itsDependencies : Sequence(String) = thisModule.
106             getList(i.getAttribute('depends'));
107         itsRuleDep : Sequence(XML!Element) = XML!Element.allInstances() ->
108             select(d | d.name = 'rule'
109                 and itsDependencies ->
110                     includes( d.getAttribute('name')));
111         itsFileDep : Sequence(String) = itsDependencies ->
112             select(d | allRules -> select(e|e.getAttribute('name')=d)->
113                 isEmpty() ) ;
114     }
115     to o : Make!Rule(
116         name <- i.getAttribute('name'),
117         dependencies <- Sequence{makeRuleDep,makeFileDep},
118         shellLines <- i.children ->
119             select(d | d.ocIsKindOf(XML!Element) and d.name = 'shellLine')
120     ),
```

```
121     makeRuleDep : distinct Make!RuleDep foreach(dep in itsRuleDep)(
122         ruledep <- dep
123     ),
124     makeFileDep : distinct Make!FileDep foreach(depFile in itsFileDep)(
125         name <- depFile
126     )
127 }
128
129 rule Macro{
130     from i : XML!Element(
131         i.name = 'macro'
132     )
133     to o : Make!Macro(
134         name <- i.getAttribute('name'),
135         value <- i.getAttribute('value')
136     )
137 }
138
139 rule ShellLine{
140     from i : XML!Element(
141         i.name = 'shellLine'
142     )
143     to o : Make!ShellLine(
144         command <- i.getShellLine(),
145         display <- not (i.getText().substring(1,1)='@')
146     )
147 }
```


V. Make2Ant.atl file

```
1  module Make2Ant;
2  create OUT : Ant from IN : Make;
3
4  -- rule Makefile2Project: its the 'main' rule.
5  -- This rule generates the Project element.
6  -- Its attributes are the name and the description of the makefile.
7  -- It contains properties, targets of the makefile.
8  -- It defines also the target called by default.
9  rule Makefile2Project{
10     from
11         m : Make!Makefile
12     to
13         a : Ant!Project(
14             name      <- m.name,
15             description <- m.comment.text,
16             properties <- m.elements ->
17                 select(c | c.oclIsKindOf(Make!Macro)),
18             targets   <- m.elements ->
19                 select(c | c.oclIsKindOf(Make!Rule)),
20             default    <- m.elements ->
21                 select(c | c.oclIsKindOf(Make!Rule)) -> first()
22         )
23     }
24
25  -- rule Macro2Property:
26  -- This rule generates a Property.
27  -- Its attributes are the name and the value of the Macro.
28  rule Macro2Property{
29     from
30         m : Make!Macro
31     to
32         a : Ant!PropertyValue(
33             name <- m.name,
34             value <- m.value
35         )
36     }
37
38  -- rule Rule2Target:
39  -- This rule generates a Target.
40  -- Its attribute is the name of the rule.
41  -- It contains tasks.
42  -- It can be dependent of others targets.
43  rule Rule2Target{
44     from
45         m : Make!Rule
46     to
47         a:Ant!Target(
48             name      <- m.name,
49             tasks     <- m.shellLines,
50             depends   <- m.dependencies ->
51                 select(e | e.oclIsKindOf(Make!RuleDep)) ->
52                 collect(e | e.ruledep)
53         )
54     }
55
56  -- rule ShellLine2Task_Display
57  -- This rule is started when the attribute display is true.
58  -- This rule generate a Task Echo and a Task Exec.
```

```
59 rule ShellLine2Task_Display{
60   from
61     m : Make!ShellLine(
62       m.display
63     )
64   to
65     e:Ant!Echo(
66       message <- m.command,
67       target <- m.ruleShellLine
68     ),
69     x:Ant!Exec(
70       executable <- m.command,
71       target <- m.ruleShellLine
72     )
73 }
74
75 -- rule ShellLine2Task_Display
76 -- This rule is started when the attribute display is false.
77 -- This rule generate only a Task Exec.
78 rule ShellLine2Task_NoDisplay{
79   from
80     m : Make!ShellLine(
81       not m.display
82     )
83   to
84     x:Ant!Exec(
85       executable <- m.command,
86       target <- m.ruleShellLine
87     )
88 }
```

VI. Ant2XML.atl file


```
1  module Ant2XML;
2  create OUT : XML from IN : Ant;
3
4  -- concatene a list of String
5  -- the elements are separated by a comma
6  helper def: concat(list : Sequence(String)) : String =
7      list -> asSet() -> iterate(element ;acc : String = '' |
8          acc +
9              if acc = ''
10             then element
11             else ',' + element
12             endif);
13
14 -- rule for a project having a description
15 rule Project2Root{
16     from i : Ant!Project(
17         if i.description.oclIsUndefined()
18         then false
19         else not(i.description='')
20         endif
21     )
22     to o : XML!Root(
23         name <- 'project',
24         children <- Sequence {itsName,itsDescription,itsBasedir,
25                             itsDefaultTarget,i.properties,
26                             i.path,i.taskdef,i.targets}
27     ),
28     itsName : XML!Attribute(
29         name <- 'name',
30         value <- i.name
31     ),
32     itsDescription : XML!Element(
33         name <- 'description',
34         children <- textText
35     ),
36     textText : XML!Text(
37         value <- i.description
38     ),
39     itsBasedir : XML!Attribute(
40         name <- 'basedir',
41         value <- i.basedir
42     ),
43     itsDefaultTarget : XML!Attribute(
44         name <- 'default',
45         value <- i.default.name
46     )
47 }
48
49 -- rule for a project without description
50 rule Project2RootWithoutDescription{
51     from i : Ant!Project(
52         if i.description.oclIsUndefined()
53         then true
54         else i.description=''
55         endif
56     )
57     to o : XML!Root(
58         name <- 'project',
```

```
59     children <- Sequence {itsName,itsBasedir,
60                               itsDefaultTarget,i.properties,
61                               i.path,i.taskdef,i.targets}
62   },
63   itsName : XML!Attribute(
64     name <- 'name',
65     value <- i.name
66   ),
67   itsBasedir : XML!Attribute(
68     name <- 'basedir',
69     value <- i.basedir
70   ),
71   itsDefaultTarget : XML!Attribute(
72     name <- 'default',
73     value <- i.default.name
74   )
75 }
76
77 -- properties
78
79 rule PropertyValue{
80   from i : Ant!PropertyValue
81   to o : XML!Element(
82     name <- 'property',
83     children <- Sequence{propertyName2,propertyValue}
84   ),
85   propertyName2 : XML!Attribute(
86     name <- 'name',
87     value <- i.name
88   ),
89   propertyValue : XML!Attribute(
90     name <- 'value',
91     value <- i.value
92   )
93 }
94
95 -- ...
96
97 -- target
98 rule TargetWithDescription{
99   from i : Ant!Target(
100     if i.description.oclIsUndefined()
101     then false
102     else not (i.description='')
103     endif
104   )
105   to o : XML!Element(
106     name <- 'target',
107     children <- Sequence{nameAttribute,descriptionElement,
108                           dependsAttribute,i.tasks}
109   ),
110   nameAttribute : XML!Attribute(
111     name <- 'name',
112     value <- i.name
113   ),
114   descriptionElement : XML!Element(
115     name <- 'description',
116     children <- descriptionText
117   ),
118   descriptionText : XML!Text(
119     value <- i.description
120   ),
```

```
121     dependsAttribute : XML!Attribute(  
122         name <- 'depends',  
123         value <- thisModule.concat(i.depends -> collect(e|e.name))  
124     )  
125 }  
126  
127 rule TargetWithoutDescription{  
128     from i : Ant!Target(  
129         if i.description.oclIsUndefined()  
130         then true  
131         else i.description=''  
132         endif  
133     )  
134     to o : XML!Element(  
135         name <- 'target',  
136         children <- Sequence{nameAttribute,dependsAttribute,i.tasks}  
137     ),  
138     nameAttribute : XML!Attribute(  
139         name <- 'name',  
140         value <- i.name  
141     ),  
142     dependsAttribute : XML!Attribute(  
143         name <- 'depends',  
144         value <- thisModule.concat(i.depends -> collect(e|e.name))  
145     )  
146 }  
147  
148 -- tasks  
149 -- ...  
150  
151  
152 -- pre-defined tasks  
153 -- ...  
154  
155 rule Exec{  
156     from i : Ant!Exec  
157     to o : XML!Element(  
158         name <- 'exec',  
159         children <- execAttribute  
160     ),  
161     execAttribute : XML!Attribute(  
162         name <- 'executable',  
163         value <- i.executable  
164     )  
165 }  
166  
167 rule Echo{  
168     from i : Ant!Echo  
169     to o : XML!Element(  
170         name <- 'echo',  
171         children <- echoAttribute  
172     ),  
173     echoAttribute : XML!Attribute(  
174         name <- 'message',  
175         value <- i.message  
176     )  
177 }  
178 -- ...  
179 -- it exists others rules in this file, but they are never called for this --  
180 project.
```

VII. XML2Text.atl file

```
1  query XML2Text = XML!Root.allInstances()
2      ->asSequence()
3      ->first().toString2('').writeTo('MyDirectory\\project.xml');
4
5
6  helper context XML!Element def: toString2(indent : String) : String =
7      let na : Sequence(XML!Node) =
8          self.children->select(e | not e.oclIsKindOf(XML!Attribute)) in
9      let a : Sequence(XML!Node) =
10         self.children->select(e | e.oclIsKindOf(XML!Attribute)) in
11         indent + '<' + self.name +
12         a->iterate(e; acc : String = '' |
13             acc + ' ' + e.toString2()
14         ) +
15         if na->size() > 0 then
16             '>'
17             + na->iterate(e; acc : String = '' |
18                 acc +
19                 if e.oclIsKindOf(XML!Text) then
20                     ''
21                 else
22                     '\r\n'
23                 endif
24                 + e.toString2(indent + ' ')
25             ) +
26             if na->first().oclIsKindOf(XML!Text) then
27                 '</' + self.name + '>'
28             else
29                 '\r\n' + indent + '</' + self.name + '>'
30             endif
31         else
32             '>'
33         endif;
34
35
36  helper context XML!Attribute def: toString2() : String =
37      if (self.value.oclIsUndefined()) or (self.value='')
38      then ''
39      else self.name + '=' + self.value + '"'
40      endif;
41  --self.name + '=' + self.value + '"';
42
43  helper context XML!Text def: toString2() : String =
44      self.value;
```

	ATL TRANSFORMATION EXAMPLE	
	Make to Ant	Date 05/08/2005

References

- [1] Make Description. <http://www.gnu.org/software/make/>
- [2] Overview of Ant. <http://ant.apache.org/manual/>
- [3] KM3: Kernel MetaMetaModel. Available at <http://dev.eclipse.org/viewcvs/indextech.cgi/~checkout~/gmt-home/doc/atl/index.html>.